Math 277
Ordinary Differential Equations
Fall 2005

Dr. Constant J. Goutziers
Department of Mathematics, Computer Science and Statistics
goutzicj@oneonta.edu

Lesson 14

Cauchy-Euler Equations

Initializations

```
> restart;
  with(PDEtools):
```

14.1 Homogeneous Cauchy-Euler Equations.

For constants a, b, and c, a differential equation of the form

$$ax^2\frac{d^2y}{dx^2}+bx\frac{dy}{dx}+cy=0$$

is known as a homogeneous Cauchy-Euler equation. It can be transformed into a homogeneous linear equation with constant coefficients by using the transformation

$$x=e^t$$

Observe

$$\frac{dy}{dx}=\frac{\frac{dy}{dt}}{\frac{dx}{dt}}=\frac{\frac{dy}{dt}}{e^t}=e^{-t}\frac{dy}{dt}$$

and

$$\frac{d^2y}{dx^2}=\frac{\frac{d}{dt}\frac{dy}{dx}}{\frac{dx}{dt}}=e^{-t}\frac{d}{dt}\frac{dy}{dx}=e^{-t}\frac{d}{dt}\left[e^{-t}\frac{dy}{dt}\right]=e^{-t}\left[-e^{-t}\frac{dy}{dt}+e^{-t}\frac{d^2y}{dt^2}\right]=e^{-2t}\left[-\frac{dy}{dt}+\frac{d^2y}{dt^2}\right]$$

Substitution of these results, and $x=\mathbf{e}^t$, in the differential equation

$$ax^2\frac{d^2y}{dx^2}+bx\frac{dy}{dx}+cy=0$$

yields

$$ae^{2t}\left\{e^{-2t}\left[-\frac{dy}{dt}+\frac{d^2y}{dt^2}\right]\right\}+be^t\left\{e^{-t}\frac{dy}{dt}\right\}+cy=0$$

Elementary simplification of this equation gives

$$a\frac{d^2y}{dt^2}+(b-a)\frac{dy}{dt}+cy=0$$

Maple will quickly collaborate this result.

```
> deq_x:=a*x^2*diff(y(x), x$2)+b*x*diff(y(x), x)+c*y(x)=0;
```

$$deq\_x := a\,x^2\left(\frac{d^2}{dx^2}\mathrm{y}(x)\right)+b\,x\left(\frac{d}{dx}\mathrm{y}(x)\right)+c\,\mathrm{y}(x)=0$$

```
> tr:=x=exp(t);
  deq_t:=collect(dchange(tr, deq_x), diff);
```

$$tr := x=\mathbf{e}^t$$

$$deq\_t := (-a+b)\left(\frac{d}{dt}\mathrm{y}(t)\right) + a\left(\frac{d^2}{dt^2}\mathrm{y}(t)\right) + c\,\mathrm{y}(t) = 0$$

Examples

Example 14.1.1

Solve the initial value problem

$$2x^2y'' + 5xy' + 3y = 0,\; y(1) = 2,\; y'(1) = -3$$

```
> deq_x:=2*x^2*diff(y(x), x$2)+5*x*diff(y(x), x)+3*y(x)=0;
```

$$deq\_x := 2x^2\left(\frac{d^2}{dx^2}\mathrm{y}(x)\right) + 5x\left(\frac{d}{dx}\mathrm{y}(x)\right) + 3\,\mathrm{y}(x) = 0$$

Using the transformation $x = \mathbf{e}^t$, we transform the equation into a linear homogeneous equation with constant coefficients.

```
> tr:=x=exp(t);
  deq_t:=collect(dchange(tr, deq_x), diff);
```

$$tr := x = \mathbf{e}^t$$

$$deq\_t := 3\left(\frac{d}{dt}\mathrm{y}(t)\right) + 2\left(\frac{d^2}{dt^2}\mathrm{y}(t)\right) + 3\,\mathrm{y}(t) = 0$$

Compute the auxiliary equation.

```
> aux_eq:=simplify(eval(subs(y(t)=exp(r*t), deq_t))/exp(r*t));
```

$$aux\_eq := 3r + 2r^2 + 3 = 0$$

Find the eigenvalues.

```
> evals:=solve(aux_eq, r);
```

$$evals := -\frac{3}{4} + \frac{1}{4}I\sqrt{15},\, -\frac{3}{4} - \frac{1}{4}I\sqrt{15}$$

Generate the corresponding solutions.

```
> Z[1]:=exp(Re(evals[1])*t)*cos(Im(evals[1])*t);
  Z[2]:=exp(Re(evals[1])*t)*sin(Im(evals[1])*t);
```

$$Z_1 := \mathbf{e}^{\left(-\frac{3t}{4}\right)}\cos\left(\frac{\sqrt{15}\,t}{4}\right)$$

$$Z_2 := \mathbf{e}^{\left(-\frac{3t}{4}\right)}\sin\left(\frac{\sqrt{15}\,t}{4}\right)$$

Create the general solution for the equation "deq_t"

```
> z:=add(c[k]*Z[k], k=1..2);
```

$$z := c_1\,\mathbf{e}^{\left(-\frac{3t}{4}\right)}\cos\left(\frac{\sqrt{15}\,t}{4}\right) + c_2\,\mathbf{e}^{\left(-\frac{3t}{4}\right)}\sin\left(\frac{\sqrt{15}\,t}{4}\right)$$

Finally, we need to transform this expression in terms of the variable x. Recall that $x = \mathbf{e}^t$, so $t = \ln(x)$.

```
> tr_inv:=isolate(tr, t);
```

$$tr\_inv := t = \ln(x)$$

The general solution of the equation

$$2x^2y'' + 5xy' + 3y = 0$$

is therefore given by

```
> gensol:=expand(subs(tr_inv, z));
```

$$gensol := \frac{c_1\cos\left(\frac{1}{4}\sqrt{15}\,\ln(x)\right)}{x^{(3/4)}} + \frac{c_2\sin\left(\frac{1}{4}\sqrt{15}\,\ln(x)\right)}{x^{(3/4)}}$$

When implementing the initial conditions, it is advantageous to convert the expression above into Maple function.

```
> yg:=unapply(gensol, x);
```

$$yg := x \to \frac{c_1 \cos\left(\frac{1}{4}\sqrt{15}\ \ln(x)\right)}{x^{(3/4)}} + \frac{c_2 \sin\left(\frac{1}{4}\sqrt{15}\ \ln(x)\right)}{x^{(3/4)}}$$

```
> eq1:=yg(1)=2;
  eq2:=D(yg)(1)=-3;
  val_c:=solve({eq1, eq2}, {c[1], c[2]});
```

$$eq1 := c_1 = 2$$

$$eq2 := -\frac{3}{4} c_1 + \frac{1}{4} c_2 \sqrt{15} = -3$$

$$val\_c := \{c_1 = 2, c_2 = -\frac{2\sqrt{15}}{5}\}$$

```
> solution:=subs(val_c, yg(x));
```

$$solution := \frac{2 \cos\left(\frac{1}{4}\sqrt{15}\ \ln(x)\right)}{x^{(3/4)}} - \frac{2}{5}\frac{\sqrt{15}\ \sin\left(\frac{1}{4}\sqrt{15}\ \ln(x)\right)}{x^{(3/4)}}$$

```
>
```